Maple-assisted proof of empirical formula for A183628

Robert Israel

30 March 2018

There are $3^6 = 729$ possible configurations for a row. Consider the 729×729 transition matrix T such that $T_{ij} = 1$ if the bottom rows of a 2×6 sub-array could be in configuration i while the top row is in configuration j (i.e. each 2×2 block has sum 4), and 0 otherwise. The following Maple code computes it. I'm encoding a row $\left[\begin{array}{cccccc} b_1 & b_2 & b_3 & b_4 & b_5 & b_6 \end{array}\right]$ as $b+1$ where $b_1 b_2 b_3 b_4 b_5 b_6$ is the base-3 representation of b. The $+1$ is needed because matrix indices start at 1 rather than 0.

```
> q:= proc(a,b) local A,B;
    A:= convert(a-1+3^6,base,3)[1..6];
    B:= convert(b-1+3^6,base,3)[1..6];
    `if`(`and`(seq(A[i]+A[i+1]+B[i]+B[i+1]=4,i=1..5)),1,0)
  end proc:
  T:= Matrix(3^6,3^6, q):
```

Thus $a(n) = u\, T^n\, v$ where u and v are row and column vectors respectively of all 1's.

```
> u:= Vector[row](3^6,1):
  v:= Vector(3^6,1):
```

To check, here are the first few entries of our sequence.

```
> V[0]:= v:
  for n from 1 to 10 do V[n]:= T . V[n-1] od:
> seq(u . V[n], n = 1 .. 10);
```

$$859, 1125, 1675, 2829, 5299, 10725, 23035, 52029, 123139, 304725 \quad (1)$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \mapsto t^6 - 3\,t^5 - 5\,t^4 + 15\,t^3 + 4\,t^2 - 12\,t \quad (2)$$

This turns out to have degree 6. Thus we will have $0 = u\, P(T)\, T^n\, v = \sum_{i=0}^{6} p_i\, a(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 6, which would hold true for any u and v. It seems that with our particular u and v we have a recurrence of order only 3, corresponding to a factor of P.

```
> factor(P(t));
```

$$t\,(t-1)\,(t-2)\,(t-3)\,(t+2)\,(t+1) \quad (3)$$

```
> Q:= unapply(expand((t-1)*(t-2)*(t-3)), t);
```

$$Q := t \mapsto t^3 - 6\,t^2 + 11\,t - 6 \quad (4)$$

The complementary factor $R(t) = \dfrac{P(t)}{Q(t)}$ has degree 3.

```
> R:= unapply(normal(P(t)/Q(t)), t);
```

$$R := t \mapsto \left(t^2 + 3\,t + 2\right) t \quad (5)$$

Now we want to show that $b(n) = u\, Q(T)\, T^n\, v = 0$ for all n. This will certainly satisfy the order-3 recurrence

$$\sum_{i=0}^{3} r_i\, b(i+n) = \sum_{i=0}^{3} r_i u\, Q(T)\, T^{n+i} v = u\, Q(T)\, R(T)\, T^n\, v = u\, P(T)\, T^n\, v = 0$$

where r_i are the coefficients of $R(t)$. To show all $b(n) = 0$ it suffices to show $b(0) =\ldots = b(2) = 0$.

```
> uQT:= u . Q(T): seq( uQT . V[n], n = 0 .. 2);
```

$$0, 0, 0 \tag{6}$$

```
> rsolve({a(n)=6*a(n-1)-11*a(n-2)+6*a(n-3), a(1)= 859,a(2)=1125,a
  (3)=1675},a(n));
```

$$124\, 2^n + 3\, 3^n + 602 \tag{7}$$

```
> gf:= sum(%*x^n,n=1..infinity,formal);
```

$$gf := \frac{-4374\, x^3 + 4029\, x^2 - 859\, x}{6\, x^3 - 11\, x^2 + 6\, x - 1} \tag{8}$$